

Functional Programming + Dependent Types \equiv Verified Linear Algebra

github.com/ryanorendorff/functional-linear-algebra-talk

Ryan Orendorff

2021



Research Scientist at Facebook Reality Labs (FRL).

- Passionate about theorem proving, programming language theory and dependently typed languages such as Agda.
- Repos and other talks can be found here:
github.com/ryanorendorff/

Disclaimer: This work is done on personal time/equipment and is not sponsored by my employer (past or present), nor is this talk related to work at any employer.

Goal of the talk: correct by construction matrices

We want to be able to enforce that a user cannot create an incorrect matrix, or use a matrix improperly.

There are a few ways things can go wrong:

- Improper sizing

```
data Matrix a = Matrix [[a]]  
testMatrix = Matrix [[1, 2, 3], [3, 4]]
```

- Improper data types

```
testMatrix = Matrix [ (Just 5) ]
```

Plus a few more surprising errors to get to later!

First step: define a type for a matrix

A matrix is often a table of numbers, which we can encode in Agda as

```
data MatrixOfNumbers (A : Set) : Set where  
    ConstructMatrixOfNumbers : List (List A) → MatrixOfNumbers A
```

which is equivalent to the Haskell

```
data MatrixOfNumbers a = ConstructMatrixOfNumbers [[a]]
```

First step: define a type for a matrix

Given this constructor we can create the following matrix.

$$M_n = \begin{bmatrix} 1 & 2 & 3 \\ 4 & 5 & 6 \end{bmatrix}$$

M_n : MatrixOfNumbers \mathbb{N} -- *Natural numbers*

M_n = ConstructMatrixOfNumbers [[1 , 2 , 3] , [4 , 5 , 6]]

Conventions used in this talk : A is a type, M_i is a matrix, $m\ n\ p\ q$ are natural numbers, and $u\ v\ x\ y$ are vectors.

What can we do with a matrix?

A matrix can be used in a few different cases:

1. Multiply a matrix with a vector (matrix-vector multiply): Mx
2. Transform a matrix to get a new matrix (transpose): $M^T x$
3. Combine matrices (matrix-matrix multiply): $M_1 * M_2$

What is matrix-vector multiplication?

Matrix-vector multiply transforms one vector into another through multiplication and addition.

$$\begin{bmatrix} 1 & 2 & 3 \\ 4 & 5 & 6 \end{bmatrix} * \begin{bmatrix} 1 \\ 2 \\ 3 \end{bmatrix} = \begin{bmatrix} (1 * 1) + (2 * 2) + (3 * 3) \\ (4 * 1) + (5 * 2) + (6 * 3) \end{bmatrix} = \begin{bmatrix} 14 \\ 32 \end{bmatrix}$$
$$M \quad * \quad x \quad = \quad y$$

Another way to think of matrix-vector multiplication: M is a *function* from vectors of size 3 to vectors of size 2. This function is sometimes called a *linear map*.

Example of a matrix as a function: identity

The identity matrix converts a vector into the same vector.

$$I * v = \begin{bmatrix} 1 & \cdots & 0 \\ \vdots & \ddots & \vdots \\ 0 & \cdots & 1 \end{bmatrix} * v = v$$

If we were to write out the identity matrix as a function, it would be the same as the identity function.

`list-identity : List A → List A`

`list-identity l = l`

Example of a matrix as a function: diag

The diagonal matrix element-wise multiplies one vector with another (written as $*^V$).

$$\text{diag}(u) * v = \begin{bmatrix} u_1 & \cdots & 0 \\ \vdots & \ddots & \vdots \\ 0 & \cdots & u_n \end{bmatrix} * v = u *^V v$$

written as a function, this would look like

```
diag : List A → (List A → List A)
```

```
diag u = λ v → u *V v
```

or alternatively as

```
diag u = λ v → zipWith _ (_*_) u v
```

-- zipWith in Agda has an extra parameter we don't need

Let's define a matrix as a function!

We can define a matrix as a function that takes a vector and returns a new vector.

```
data FunctionalMatrix (A : Set) : Set where  
  ConstructFunctionalMatrix : (List A → List A) → FunctionalMatrix A
```

Now we can construct the identity matrix as follows:

```
Mi : FunctionalMatrix A  
Mi = ConstructFunctionalMatrix (list-identity)
```

This addresses the matrix-vector ability of a matrix, what else can we tackle functionally?

We have matrix-vector multiply down, can we do more?

With our functional definition of a matrix, we can do other operations like matrix-matrix multiply.

$$(M_1 * M_2)v = M_1(M_2(v))$$

```
_·f_ : FunctionalMatrix A → List A → List A
```

```
ConstructFunctionalMatrix f ·f l = f l
```

```
apply_two_matrices : FunctionalMatrix A → FunctionalMatrix A  
                    → List A → List A
```

```
apply_two_matrices M1 M2 v = M1 ·f M2 ·f v
```

Hmm that looks a lot like composition:

```
_°f_ : FunctionalMatrix A → FunctionalMatrix A → FunctionalMatrix A  
M1 °f M2 = ConstructFunctionalMatrix (apply_two_matrices M1 M2)
```

We often need the transpose matrix at the same time

This type encapsulates the function nature of a matrix, but we often need the transpose as well.

```
data FunctionalMatrixWithTranspose (A : Set) : Set where  
  ConstructFMT : (List A → List A) -- Forward function  
                → (List A → List A) -- Transpose function  
                → FunctionalMatrixWithTranspose A
```

We can now define the identity matrix with the transpose matrix function, which is also the identity.

```
Mi,t : FunctionalMatrixWithTranspose A  
Mi,t = ConstructFMT (list-identity) (list-identity)
```

Intuition check : convert a functional matrix into a number matrix

For the rest of linear algebra to work, we should always be able to define an equivalent functions using only multiplication and addition.

For example our original identity function

```
identity' : List A → List A
```

```
identity' v = v
```

could be written as

```
identity' : List A → List A
```

```
identity' v = replicate (len v) 1 *v v
```

where **replicate** creates a list of 1s and ***v** multiplies element-wise.

Is `FunctionalMatrixWithTranspose` “correct by construction”?

Our original goal was

Correct by construction linear algebra; equivalent to \mathbb{R} matrices

Is this true for `FunctionalMatrixWithTranspose`?

`f1 : List ℕ → List ℕ`

`f1 v = []`

`Mr : FunctionalMatrixWithTranspose ℕ`

`Mr = ConstructFMT f1 f1`

Hmm, intuition check: can we write `Mr` as a matrix of numbers?

$$M_r = \begin{bmatrix} \text{throw away data} \\ \text{throw away data} \end{bmatrix}$$

Encoding the length of the vector in the type

Agda allows us to specify what the length of a vector as part of the type.¹

```
v : Vec ℕ 3
```

```
v = [ 1 , 2 , 3 ]v
```

If we try to create a vector of the wrong length, Agda will tell us.

```
v2 : Vec ℕ 2
```

```
v2 = v
```

```
-- Get the following error: 3 ≠ 2 of type ℕ
```

`Vec` is a *dependent type* because its type *depends on a value*.

¹This is a bit of a misnomer; the difference between term and type is muddled in many dependently typed languages.

Use functions on Vec to ensure that the shapes match

We can define a matrix type where the shapes are preserved.

```
data SizedMatrix (A : Set) (m n : ℕ) : Set where  
  ConstructSizedMatrix : (Vec A n → Vec A m) -- Forward function  
                        → (Vec A m → Vec A n) -- Transpose function  
                        → SizedMatrix A m n
```

Previously this would be done with a runtime check.

In Haskell, we would write this as

```
data SizedMatrix (A :: *) (m :: Nat) (n :: Nat) where  
  ConstructSizedMatrix :: (KnownNat m, KnownNat n)  
                        ⇒ (Vec A n → Vec A m) -- Forward function  
                        → (Vec A m → Vec A n) -- Transpose function  
                        → SizedMatrix A m n
```


Use functions on Vec to ensure that the shapes match

We can define a matrix type where the shapes are preserved.

```
data SizedMatrix (A : Set) (m n : ℕ) : Set where  
  ConstructSizedMatrix : (Vec A n → Vec A m) -- Forward function  
                        → (Vec A m → Vec A n) -- Transpose function  
                        → SizedMatrix A m n
```

We can now define our identity matrix again.

```
id : (A : Set) → A → A
```

```
Mi,s : SizedMatrix A n n
```

```
Mi,s = ConstructSizedMatrix id id -- id : Vec A n → Vec A n
```

Intuition check: can we encode matrices that are not possible to write out?

Our original goal was

Correct by construction linear algebra; equivalent to \mathbb{R} matrices

We could write a matrix for handling playing cards.

data Card : Set **where**

♠ ♣ ♥ ♦ : Card

M_{\spadesuit} : SizedMatrix Card n n

$M_{\spadesuit} = \text{ConstructSizedMatrix } (\lambda v \rightarrow \text{replicate } \clubsuit) (\lambda v \rightarrow \text{replicate } \heartsuit)$

If we wanted to convert this to multiplication and addition only....

$$M_{\spadesuit} = \begin{bmatrix} \clubsuit & \heartsuit \\ \diamond & \spadesuit \end{bmatrix}$$

Matrices cannot contain just anything! The elements have to be able to be added/multiplied.

Matrices are defined over fields

To check our intuition we have been trying to determine if our function could be written using multiplication and addition. Formally, this is equivalent to saying the elements of a matrix form a *Field*.

```
record Field (A : Set) : Set where
```

```
field
```

```
  _+_ : A → A → A -- 3 + 4
```

```
  _*_ : A → A → A -- 3 * 4
```

```
  _- : A → A -- + inverse, - 4
```

```
  ^-1 : A → A -- * inverse, 4 ^-1
```

```
  0f : A -- Identity of _+_, 4 + 0f = 4
```

```
  1f : A -- Identity of _*_ , 4 * 1f = 4
```

We can define matrices that operate on Fields only

We can restrict our `A` type to having a defined version of `+` and `*`.

```
data SizedFieldMatrix (A : Set) { F : Field A } (m n : ℕ) : Set where  
  ConstructSFM : (Vec A n → Vec A m) -- Forward function  
                → (Vec A m → Vec A n) -- Transpose function  
                → SizedFieldMatrix A m n
```

in Haskell this would be written as

```
data SizedFieldMatrix A (m :: Nat) (n :: Nat) where  
  ConstructSFM :: (KnownNat m, KnownNat n, Field A)  
                => (Vec A n → Vec A m) -- Forward function  
                → (Vec A m → Vec A n) -- Transpose function  
                → SizedFieldMatrix A m n
```

We can define matrices that operate on Fields only

We can restrict our A type to having a defined version of $+$ and $*$.

```
data SizedFieldMatrix (A : Set) { F : Field A } (m n : ℕ) : Set where  
  ConstructSFM : (Vec A n → Vec A m) -- Forward function  
                → (Vec A m → Vec A n) -- Transpose function  
                → SizedFieldMatrix A m n
```

The card example can no longer be constructed, but the identity matrix still can be constructed.

*-- + and * must be defined on A*

```
Msfi : { F : Field A } → SizedFieldMatrix A n n  
Msfi = ConstructSFM id id
```

Intuition check: can we encode matrices that are not possible to write out?

Our original goal was

Correct by construction linear algebra; equivalent to \mathbb{R} matrices

$M_{\text{rep}} : \{ F : \text{Field } A \} \rightarrow \text{SizedFieldMatrix } A \text{ } m \text{ } n$

$M_{\text{rep}} = \text{ConstructSFM } (\lambda v \rightarrow \text{replicate } 1^f) (\lambda v \rightarrow \text{replicate } 1^f)$

Using only multiplication and addition does not allow us to *produce constant outputs for any input*.

Matrices are linear functions

Matrices have the following properties that we'd like to preserve:

- Linearity: $M(u +^V v) = M(u) +^V M(v)$
- Homogeneity: $M(c \circ^V v) = c \circ^V M(v)$

Currently we could define a matrix like so, which has neither property.

```
_ : { F : Field A } → SizedFieldMatrix A n n  
_ = ConstructSFM (λ v → replicate 1 f) (λ v → replicate 1 f)
```

- Linearity: $f(u +^V v) = \vec{1} \neq f(u) +^V f(v) = \vec{1} +^V \vec{1} = \vec{2}$
- Homogeneity: $f(c \circ^V v) = \vec{1} \neq c \circ^V f(v) = c \circ^V \vec{1} = \vec{c}$

How do we ensure that our functions are linear?

For our matrices to make sense, we need the functions that are used for the forward and transpose functions to be linear functions.

-- A linear function (aka a linear map)

```
record _~_ {A : Set} ⌋ F : Field A ⌋ (m n : ℕ) : Set where  
  field  
    f : (Vec A m → Vec A n)  
    f[u+v]≡f[u]+f[v] : (u v : Vec A m) → f (u +v v) ≡ f u +v f v  
    f[c*v]≡c*f[v] : (c : A) → (v : Vec A m) → f (c ◦v v) ≡ c ◦v (f v)
```

with this we could define our matrices using linear functions.

```
data LinearMatrix {A : Set} ⌋ F : Field A ⌋ (m n : ℕ) : Set where  
  ConstructLinearMatrix : (n ~ m) → (m ~ n) → LinearMatrix m n
```


What is this \equiv thing?

The \equiv sign means that two things are equal² in the sense that the left and the right side are written with the same order of constructors³.

The definition of \equiv is

```
data _ $\equiv$ _ (x : A) : A  $\rightarrow$  Set where  
  refl : x  $\equiv$  x
```

²Homogenously

³Their normal forms are equivalent

Fields must follow some properties on top of defining + and *

Fields define more than just + and *; a field has some properties.

$$+-\text{assoc} \quad : (a \ b \ c : A) \rightarrow a + (b + c) \equiv (a + b) + c$$

$$+-\text{comm} \quad : (a \ b : A) \rightarrow a + b \equiv b + a$$

$$+\mathbf{0^f} \quad : (a : A) \rightarrow a + \mathbf{0^f} \equiv a$$

$$+-\text{inv} \quad : (a : A) \rightarrow (-\ a) + a \equiv \mathbf{0^f}$$

$$*- \text{assoc} \quad : (a \ b \ c : A) \rightarrow a * (b * c) \equiv (a * b) * c$$

$$*- \text{comm} \quad : (a \ b : A) \rightarrow a * b \equiv b * a$$

$$*\mathbf{1^f} \quad : (a : A) \rightarrow a * \mathbf{1^f} \equiv a$$

$$*- \text{inv} \quad : (a : A) \rightarrow (a \neq \mathbf{0^f}) \rightarrow (a^{-1}) * a \equiv \mathbf{1^f}$$

$$*- \text{distr} + \quad : (a \ b \ c : A) \rightarrow a * (b + c) \equiv (a * b) + (a * c)$$

How do we generate these proofs? By composing them!

We can prove $(b + 0^f) * 1^f \equiv b$ from simpler (axiomatic) statements.

```
new_proof : (b : A) → (b + 0f) * 1f ≡ b
```

```
new_proof b = begin
```

```
  (b + 0f) * 1f
```

```
≡⟨ *1f (b + 0f) ⟩ -- *1f : (a : A) → a * 1f ≡ a
```

```
  b + 0f
```

```
≡⟨ +0f b ⟩ -- +0f : (a : A) → a + 0f ≡ a
```

```
  b ■
```

Proving that the identity function is linear

The linear identity function is simple

```
id1 : { F : Field A } → n → n
```

```
id1 = record
```

```
  { f = id -- Vec A n → Vec A n
```

```
  ; f[u+v]≡f[u]+f[v] = λ u v → refl -- id (u +v v) ≡ id u +v id v
  -- Reflexive because Agda can apply all three `id` and conclude
  -- u +v v ≡ u +v v
```

```
  ; f[c*v]≡c*f[v] = λ c v → refl -- id (c ∘v v) ≡ c ∘v id v
  -- Reflexivity for same reason as above proof.
  }
```

Proving that the **diag** function is linear

Now let's try to define the **diag** function as a linear function

```
diag1 : { F : Field A } → Vec A n → n → n
```

```
diag1 d = record
```

```
  { f = d *V _
```

```
  -- *V-distr-+V : d *V (u +V v) ≡ d *V u +V d *V v
```

```
  ; f[u+v]≡f[u]+f[v] = λ u v → *V-distr-+V d u v
```

```
  -- *V◦V≡◦V*V : d *V (c ◦V v) ≡ c ◦V (d *V v)
```

```
  ; f[c*v]≡c*f[v] = λ c v → *V◦V≡◦V*V c d v
```

```
  }
```

Let's go through the linearity proof for **diag**

To prove linearity for **diag**, let's step through the proof.

$$\begin{aligned} & *^V\text{-distr-}+^V : (d \ u \ v : \text{Vec } A \ n) \\ & \quad \rightarrow d \ *^V (u +^V v) \equiv d \ *^V u +^V d \ *^V v \\ & *^V\text{-distr-}+^V []^V []^V []^V = \text{refl} \\ & *^V\text{-distr-}+^V (d_\theta ::^V d_r) (u_\theta ::^V u_r) (v_\theta ::^V v_r) = \text{begin} \\ & \quad (d_\theta * (u_\theta + v_\theta)) ::^V (d_r *^V (u_r +^V v_r)) \\ & \equiv \langle \text{cong } ((d_\theta * (u_\theta + v_\theta)) ::^V _) (*^V\text{-distr-}+^V d_r \ u_r \ v_r) \rangle \\ & \quad (d_\theta * (u_\theta + v_\theta)) ::^V (d_r *^V u_r +^V d_r *^V v_r) \\ & \equiv \langle \text{cong } (_ ::^V (d_r *^V u_r +^V d_r *^V v_r)) (*^V\text{-distr-}+^V d_\theta \ u_\theta \ v_\theta) \rangle \\ & \quad (d_\theta * u_\theta + d_\theta * v_\theta) ::^V (d_r *^V u_r +^V d_r *^V v_r) \\ & \equiv \langle \rangle \\ & \quad (d_\theta ::^V d_r) *^V (u_\theta ::^V u_r) +^V (d_\theta ::^V d_r) *^V (v_\theta ::^V v_r) \blacksquare \end{aligned}$$

How we can finally define LinearMatrix!

We can finally define a matrix made up of linear functions.

```
data LinearMatrix {A : Set} {F : Field A} (m n : ℕ) : Set where
```

```
  ConstructLinearMatrix : (n → m) → (m → n) → LinearMatrix m n
```

```
id-linear : {F : Field A} → LinearMatrix n n
```

```
id-linear = ConstructLinearMatrix id₁ id₁
```

Have we reached “Correct by construction linear algebra”?

Does the transpose match?

Say we defined a matrix as so

```
Mno : { F : Field A } → LinearMatrix n n  
Mno = ConstructLinearMatrix (id1) (diag1 (replicate 0f))
```

We have mixed up the forward/transpose pairing.

$$I = I^T$$

$$\text{diag}(v) = \text{diag}(v)^T$$

To solve this problem, we can show that for forward function M and transpose function M^T that the following property holds.

$$\forall xy. \langle x, My \rangle = \langle y, M^T x \rangle$$

$$\langle a, b \rangle = \text{sum}(a *^V b) = \sum_i^n a_i * b_i$$

Finally we reach our goal of correct by construction matrices!

If we require the user to prove the inner product property, we can *finally* create a “correct by construction” functional matrix.

```
data Mat_×_ {A : Set} {F : Field A} (m n : ℕ) : Set where  
  [[_,_,_]] : (M : n ~ m)  
    → (MT : m ~ n)  
    → (p : (x : Vec A m) → (y : Vec A n)  
        → ⟨ x , M ·1 y ⟩ ≡ ⟨ y , MT ·1 x ⟩ )  
    → Mat m × n
```

where the inner product ($\langle \rangle$) is defined as

```
⟨_,_⟩ : {F : Field A} → Vec A n → Vec A n → A  
⟨ x , y ⟩ = sum (x *v y)  
-- Sum of the element-wise multiply of two vectors
```

The final identity functional matrix

With this, we can finally define the identity matrix in a way that is not possible to make an error.

$M^I : \{ F : \text{Field } A \} \rightarrow \text{Mat } n \times n$

$M^I = \llbracket \text{id}_1, \text{id}_1, \text{id-transpose} \rrbracket$

where

$\text{id-transpose} : \{ F : \text{Field } A \} (x\ y : \text{Vec } A\ n)$
 $\rightarrow \langle x, \text{id } y \rangle \equiv \langle y, \text{id } x \rangle$

$\text{id-transpose } x\ y = \text{begin}$
 $\langle x, \text{id } y \rangle \equiv \langle \rangle$
 $\langle x, y \rangle \equiv \langle \langle \rangle\text{-comm } x\ y \rangle$
 $\langle y, x \rangle \equiv \langle \rangle$
 $\langle y, \text{id } x \rangle \blacksquare$

What can we do with a matrix

We can do a few things with a matrix:

1. Multiply the matrix with a vector (matrix-vector multiply): Mx
2. Transform the matrix to get a new matrix (transpose): $M^T x$
3. Combine matrices (matrix-matrix multiply): $M_1 * M_2$

We have not done matrix-matrix multiplication, can we implement it with our new definition?

Implementing matrix-matrix multiply on functional matrices

Previously, we were able to define matrix-matrix multiplication

$$M_1 * M_2$$

using function composition

```
apply_two_matrices : FunctionalMatrix A → FunctionalMatrix A  
                    → List A → List A  
apply_two_matrices F G v = F .f G .f v
```

```
_ .f _ : FunctionalMatrix A → FunctionalMatrix A → FunctionalMatrix A  
F .f G = ConstructFunctionalMatrix (apply_two_matrices F G)
```

We can do the same with our new definition, by performing composition of the linear functions.

We are going to need a few functions to implement matrix multiply. First we define a helper function to extract the forward function.

$$\text{Mat-to--} : \{ F : \text{Field } A \} \rightarrow \text{Mat } m \times n \rightarrow n \sim m$$
$$\text{Mat-to-- } \llbracket f, t, p \rrbracket = f$$

And then we can extract the transpose by generating the transpose matrix and running `Mat-to--`.

$$_T : \{ F : \text{Field } A \} \rightarrow \text{Mat } m \times n \rightarrow \text{Mat } n \times m$$
$$\llbracket f, a, p \rrbracket^T = \llbracket a, f, (\lambda x y \rightarrow \text{sym } (p \ y \ x)) \rrbracket$$

And we need to compose linear functions to compose the functions in matrices.

```
_◦¹_ : { F : Field A } → n → p → m → n → m → p
g ◦¹ h = record {
  f = λ v → g .¹ (h .¹ v)
; f[u+v]≡f[u]+f[v] = Homework
; f[c*v]≡c*f[v] = Homework }
```

Defining matrix-matrix multiply

We can now define matrix-matrix multiply.

$$\begin{aligned}\text{forward} &: M_1 * M_2 \\ \text{transpose} &: (M_1 * M_2)^T = M_2^T * M_1^T\end{aligned}$$

Which we can directly encode in Agda.

```
_*M_ : {F : Field A} → Mat m × n → Mat n × p → Mat m × p
M₂ *M M₁ =
  [ (Mat-to-- M₂) ∘¹ (Mat-to-- M₁)
  , (Mat-to-- (M₁ ⁀)) ∘¹ (Mat-to-- (M₂ ⁀))
  , MoreHomework!
  ]
```

What do we have so far with this encoding?

We have gained some nice benefits by moving to a proven type constructor.

- We can define a performant, functional version of matrix algebra.
- We can guarantee that our implementation is correct.
- We can use equational reasoning to prove two implementations are equivalent.

But there is a cost. For one file in the library that implements this idea, out of 213 lines of code:


24 lines are function definitions (11.3% of the code). *Everything else is a proof, a type signature, or an import/control statement.*

Algorithms using Linear Algebra

What are the benefits of a functional approach to linear algebra?


We gain a few benefits from using functions directly.

- Write out the model for a process in a more direct manner.
- Speed and time benefits.



Magnetic Particle Imaging reconstructs images using functional matrices

A model of how the device (left) generates signals from the sample (rat, right) is encoded as “matrix-free” functions in Python using PyOp, the python implementation of this idea.



Matrix-free methods enable significant time and space savings


We get a sizeable improvement in image reconstruction performance using a matrix-free method.

	Metric	Matrix	Matrix-Free	Improvement
	Space	150 GB	bytes	10^9x
	Time	60 min	2 min	$30x$
Use of functional concepts		No	Yes	Priceless

Magnetic Particle Imaging can be modeled using linear algebra

In MPI, we are attempting to detect where iron is within a sample.

- v : the voltages coming off of the device.
- f_e : the distribution of iron.
- M : a *function* that converts iron distributions into voltages.



Solving linear equations finds what input produces an observed result

We can write this process of converting iron distributions to voltages as

$$Mf_e = v$$

If we have the voltages v coming off the device, we want to find the iron distribution f_e that produced that signal.

To solve this problem, we want to compare how good our estimate of the input f_e is at producing the observed output v using the following function.

$$J(f_e) = f_e^T M^T M f_e - 2f_e^T M^T v$$

Taking a **step** in the right direction

One simple way to find a better f_e than some initial guess is to update f_e *in the direction of steepest descent* ∇J .

$$\begin{aligned}f_{e,i+1} &= f_{e,i} - \alpha \nabla J(f_{e,i}) \\f_{e,i+1} &= f_{e,i} - \alpha (M^T (M f_{e,i} - v))\end{aligned}$$

We can implement this in Agda as

```
step : {F : Field A}
      → (α : A) → (M : Mat m × n)
      → (v : Vec A m) → (fe : Vec A n) → Vec A n
step α M v = λ fe → fe -v α ∘v (MT ∙ (M ∙ fe -v v))
```

Gradient descent is just running **step** a bunch of times

From there, we can find the value of f_e that best matches v by iterating.

```
gradient-descent : { F : Field A }  
  → (n : ℕ)           -- Number of iterations to run  
  → (α : A)           -- Scale factor  
  → (M : Mat m × n) -- Model of system  
  → (v : Vec A m)    -- Data  
  → (fe : Vec A n)   -- Initial estimate  
  → List (Vec A n)   -- Results (farther is better)  
  
gradient-descent n α M v fe = iterate n fe (step α M v)  
-- iterate _ x f = [x, f x, f (f x), ... ]
```


We can define equivalent forms of a linear equation

We had defined our step function as

$$\text{step } \alpha \ M \ v \ f_e = f_e -^V \alpha \circ^V (M^T \cdot (M \cdot f_e -^V v))$$

is there another way to write this function?

yes!

$\text{step}' : \{ F : \text{Field } A \}$

$\rightarrow (\alpha : A) \rightarrow (M : \text{Mat } m \times n)$

$\rightarrow (v : \text{Vec } A \ m) \rightarrow (f_e : \text{Vec } A \ n) \rightarrow \text{Vec } A \ n$

$$\text{step}' \ \alpha \ M \ v \ f_e = f_e -^V \alpha \circ^V (M^T \cdot M \cdot f_e -^V M^T \cdot v)$$

Proving the two **steps** are in lock step for better performance

We can prove **step** and **step'** are the same by demonstrating that the same inputs to **step** and **step'** lead to the same result.

```
proof : { F : Field A } → (α : A)
  → (M : Mat m × n) → (v : Vec A m) → (fe : Vec A n)
  → step α M v fe ≡ step' α M v fe
proof α M v fe = begin
  fe -v α ∘v (MT · (M · fe -v v))

  -- M-distr--v : M (fe -v v) ≡ M fe -v M v
  ≡ ⟨ cong (λ z → fe -v α ∘v z) (M-distr--v (MT) (M · fe) v) ⟩
  fe -v α ∘v (MT · M · fe -v MT · v) ■
```

We can *rewrite/optimize* our program while preserving correctness.

Have we accomplished our goal?

Our original goal was

Correct by construction linear algebra; equivalent to \mathbb{R} matrices

we certainly achieved that!


- Eliminated wrong size result bugs.
- Eliminated non-linear function bugs.
- Eliminated incorrect function pairing bugs.

We could consider three levels of implementation that have different amounts of correctness built in.

- Regular functions (Python: **PyOp/PyLops**/many other libraries)
- Size-typed functions (Haskell: **convex** library)
- Linear functions (Agda: **FLA** library)


API comparison: how likely am I to use this?

Not every library is a blast to use. How do these three functional approaches stack up?




API comparison: how likely am I to use this?

Not every library is a blast to use. How do these three functional approaches stack up?




API comparison: how likely am I to use this?

Not every library is a blast to use. How do these three functional approaches stack up?



API comparison: how likely am I to use this?

Not every library is a blast to use. How do these three functional approaches stack up?



This presentation is a program!

This presentation is an Agda program! Instructions for how to load the presentation in Agda can be found at

github.com/ryanorendorff/functional-linear-algebra-talk

The full library that implements this style (without **TrustMe!**) can be found at

github.com/ryanorendorff/functional-linear-algebra

Questions?

Thanks for listening to my talk!

github.com/ryanorendorff/functional-linear-algebra-talk



Appendix

Instructions for how to run this presentation in Agda

If you have the Nix package manager installed, you can run

```
nix-shell
```

at the root of this presentation's repo and then launch emacs

```
emacs src/FormalizingLinearAlgebraAlgorithms.lagda.md
```

More information on the Agda emacs mode can be found

<https://agda.readthedocs.io/en/v2.6.1.1/tools/emacs-mode.html>. If you

use Spacemacs, the documentation for its Agda mode is

<https://www.spacemacs.org/layers/+lang/agda/README.html>.